Math 2280−1
April 22, 2011

Revisiting resonance, for the undamped forced harmonic oscillator with $\omega_0 = 1$:

$$x''(t) + x(t) = f(t) .$$

As you check in your homework this week (good review!), the variation of parameters solution to this problem with $x(0) = x_0$, $x'(0) = v_0$ is given by the formula

$$x(t) = \int_0^t f(s) \cdot \sin(t - s) ds + x_0 \cdot \cos(t) + v_0 \cdot \sin(t) .$$

For any given periodic forcing function $f(t)$ it may be hard to decide from this formula whether resonance will occur or not. Fourier series gives the precise answer, also for the more general undamped oscillator differential equation

$$x''(t) + \omega_0^2 \cdot x(t) = f(t) .$$

Exercise 1 illustrations:
1a) $f(t) = 3 \cdot \cos(t)$:

```
> restart :
  with(plots) :
> f := t→3· cos( t);
  x := t→int(f(s)·sin(t − s), s = 0 ..t);
  x(t);
  plot(x(t), t = 0 ..30, color = black);
```

$$f := t \rightarrow 3 \cos(t)$$

$$x := t \rightarrow \int_0^t f(s) \sin(t - s) \, ds$$

$$\frac{3}{2} \sin(t) \, t$$



>

1b) $f(t) = \cos(2 \cdot t) - 2 \cdot \sin(3 \cdot t)$;

```
> f := t→cos(2·t) − 2·sin(3·t);
  x := t→int(f(s)·sin(t − s), s = 0 ..t);
  x(t);
  plot(x(t), t = 0 ..30, color = black);
```

$$f := t \rightarrow \cos(2 t) - 2 \sin(3 t)$$

$$x := t \rightarrow \int_0^t f(s) \sin(t - s) \, ds$$

$$\frac{1}{3}\cos(t) - \frac{3}{4}\sin(t) - \frac{1}{3}\cos(2t) + \frac{1}{4}\sin(3t)$$



1c) $f(t) = 3\cdot\cos(t) + \sin\left(\frac{t}{2}\right)$;

```
> f := t→3·cos(t) + sin(t/2);
  x := t→int(f(s)·sin(t − s), s = 0 ..t);
  x(t);
  plot(x(t), t = 0 ..30, color = black);
```

$$f := t \to 3\cos(t) + \sin\left(\frac{1}{2}t\right)$$

$$x := t \to \int_0^t f(s)\sin(t-s)\,ds$$

$$-\frac{2}{3}\sin(t) + \frac{3}{2}\sin(t)\,t + \frac{4}{3}\sin\left(\frac{1}{2}t\right)$$



Example 1: A square wave forcing function with amplitude 1 and period 2π. This formula works until $t = 11\pi$) .

```
> plot(Heaviside(t), t =-5 ..5, color = black);

      # this is the "unit step function", equal to zero for t<0 and to 1 for
      t>0.
                           # Heaviside was an early user
```




```
> f:=t->-1+2*sum((-1)^n*Heaviside(t-n*Pi),n=-10..10);
                 #this gives the square wave for -10*Pi<t<10*Pi
```

$$f := t \to -1 + 2 \left(\sum_{n=-10}^{10} (-1)^n \operatorname{Heaviside}(t - n\pi) \right) \tag{1}$$

```
> plot(f(t),t=-30..30,color=black,title='square wave forcing function');
```




```
> y := t→int(f(s)·sin(t − s), s = 0 ..t);
  plot(y(t), t = 0 ..30, color = black, title = 'resonance?');
```

$$y := t \to \int_0^t f(s) \sin(t - s) \, ds$$



We can explain this with Fourier series. We've previously calculated Fourier series for the square wave, but we can also have Maple do this!

```
> Fouriercoeff:=proc(ff,L,a0,a,b)   #ff=function, 2L=period
  local m,  #dummy letter to index coefficients
        s;  #domain variable
  assume(m,integer);
  a0:=simplify(1/L*int(ff(s),s=-L..L)):
  a:=m->simplify(1/L*int(ff(s)*cos(Pi/L*m*s),s=-L..L)):
  b:=m->simplify(1/L*int(ff(s)*sin(Pi/L*m*s),s=-L..L)):
```

```
end:
> squarewave:=t->-1+2*Heaviside(t): #correct from -Pi to Pi
Fouriercoeff(squarewave,Pi,Asq0,Asq,Bsq):
> Asq0;
Asq(n); #should be zero since squarewave is
        #odd
Bsq(n);  #should be zero when n is even,
        #and *4/Pi)*1/n when n is odd.
```

$$0$$

$$0$$

$$-\frac{2\,(\cos(n\,\pi)-1)}{n\,\pi} \quad \textbf{(2)}$$

So our square wave is approximated by

```
> squareapprox:=t-> 4/Pi*Sum(sin((2*k-1)*t)/(2*k-1),k=1..10);
plot(squareapprox(t),t=-10..10,color=black);
```

$$squareapprox := t \to \frac{4\left(\sum_{k=1}^{10}\frac{\sin((2\,k-1)\,t)}{2\,k-1}\right)}{\pi}$$



And the piece of the particular solution corresponding to the first forcing term $\frac{4\sin(t)}{\pi}$ is $-\frac{2\,t\cos(t)}{\pi}$, and is responsible for the resonance we saw earlier:

```
> plot({y(t),-2*t*cos(t)/Pi},t=0..30,color=black);
```



Example 2: tent function, same period.

```
> tent:=t->int(f(u),u=0..t);
  plot(tent(t),t=-15..15,color=black, title='tent wave forcing function');
```

$$tent := t \to \int_0^t f(u)\, du$$




```
> z := t -> int(tent(s)*sin(t - s), s = 0..t);
  plot(z(t), t = 0..30, color = black, title = 'resonance?');
```

$$z := t \to \int_0^t tent(s)\sin(t-s)\, ds$$




```
> Fouriercoeff(tent,Pi,AT0,AT,BT):
  assume(n,integer):
  AT0;
  AT(n);
  BT(n);
```

$$\pi$$

$$\frac{2\left((-1)^{n\sim}-1\right)}{\pi\, n\sim^2}$$

$$0 \qquad (3)$$

```
> tentapprox:=t->Pi/2+sum(AT(n)*cos(n*t),n=1..20):
  plot(tentapprox(t),t=-15..15,color=black);
```



And it's the first term

$-\frac{4\cos(t)}{\pi}$ and it’s corresponding particular solution $-\frac{2\,t\sin(t)}{\pi}$ that is causing resonance:

```
> plot({z(t),-2*t/Pi*sin(t)},t=0..30,color=black);
```



Example 3: Now let’s force with a period which is not the natural one. This square wave has period 2.

```
> h:=t->-1+2*sum((-1)^n*Heaviside(t-n),n=0..30);
```

$$h := t \to -1 + 2\left(\sum_{n=0}^{30} (-1)^n \,\mathrm{Heaviside}(t-n)\right) \tag{4}$$

```
> plot(h(t),t=0..30,color=black, title=‘forcing function‘);
```




```
> w:=t->int(sin(t-tau)*h(tau),tau=0..t);
  plot(w(t),t=0..30,color=black,title=‘resonance response?‘);
```

$$w := t \to \int_0^t \sin(t-\tau)\, h(\tau)\, d\tau$$



Example 4: A square wave which does not have the natural period, but (if you think of pushing a swing) we should expect resonance!

```
> k:=t->sum(Heaviside(t-4*Pi*n)-Heaviside(t-4*Pi*n-Pi),
  n=-5..5);
```

$$k := t \rightarrow \sum_{n=-5}^{5} \left(\text{Heaviside}(t - 4\,n\,\pi) - \text{Heaviside}(t - 4\,n\,\pi - \pi) \right) \tag{5}$$

```
> plot(k(t),t=-30..30,color=black,title=`forcing function, period = ?`);
```




```
> swing:=t->int(sin(t-s)*k(s),s=0..t);
```

$$swing := t \rightarrow \int_0^t \sin(t - s)\, k(s)\; \mathrm{d}s \tag{6}$$

```
> plot(swing(t),t=0..60,color=black,title=`resonance response?`);
```



We know what happened....our function has period 4*Pi, but has n=2 non−zero Fourier coefficients:

```
> Fouriercoeff(k,2*Pi,k0,ka,kb):
  k0;
  assume(n,integer);
  ka(n);
  kb(n);
```

$$\frac{1}{2}$$

$$\frac{\sin\left(\frac{1}{2}\, n\!\sim \pi\right)}{\pi\, n\!\sim}$$

$$-\frac{-1 + \cos\left(\frac{1}{2}\, n\!\sim \pi\right)}{\pi\, n\!\sim} \tag{7}$$

```
> kapprox:=t->k0/2+sum(ka(n)*cos(1/2*n*t)+kb(n)*sin(1/2*n*t),n=1..20):
  plot(kapprox(t),t=-20..20,color=black);
```



It was the n=2 sine term $\frac{\sin(t)}{\pi}$, which caused the resonance:

```
> plot({swing(t),-t/(2*Pi)*cos(t)},t=0..30,color=black);
```